

CS 511: Assignment 5

Due: 10 December, 11:55pm

Contents

1	Assignment Objectives and Policies	1
2	Leader Election	2
3	Part I: CR-LEA	3
3.1	CR-LEA Description	3
3.2	Implementation	3
4	Part II: DKR-LEA	3
4.1	DKR-LEA Description	3
4.2	Implementation	4
4.2.1	Part 1	5
4.2.2	Part 2	6
4.3	Verification	6
5	Submission Instructions	6

1 Assignment Objectives and Policies

Get acquainted with

- simple model development in Promela;
- basic model checking in Spin;

The following Assignment Policies should be upheld:

Collaboration Policy. This homework must be done in groups of at most two. Use of the Internet is allowed, but should not include searching for existing solutions. In case it is done in pairs, only one of the members should submit through Canvas.

Under absolutely no circumstances code can be exchanged between students. Excerpts of code presented in class can be used.

Assignments from previous offerings of the course must not be re-used. Violations will be penalized appropriately.

Late Policy. Late submissions are allowed with a penalty of 2 points per hour past the deadline.

Note: This assignment draws from [?]. In particular, the description given in Sec. 2 is taken from pp. 238-239.

2 Leader Election

In current distributed systems several services are offered by some dedicated process(es) in the system. Consider, for example, address assignment and registration, query coordination in a distributed database system, clock distribution, token regeneration after token loss in a token ring network, initiation of topology updates in a mobile network, load balancing, and so forth. Usually many processes in the system are potentially capable of providing these services. However, for consistency reasons it is usually the case that at any time only one process is allowed to actually provide a given service. This process – called the “leader” – is in fact elected. Leader election is a simple form of symmetry breaking. Sometimes it suffices to elect an arbitrary process, but for other services it is important to elect the process with the best capabilities for performing that service. Here we abstract from specific capabilities and use ranking on the basis of process identities. Each process has a unique identity, and it is assumed that a total ordering exists on these identities. The idea is therefore that the higher the process’ identity, the better its capabilities.

The process of choosing a leader is known as *leader election*. A leader election algorithm (LEA):

- is decentralized;
- each node has the same local algorithm;
- each node has a unique ID and IDs form a totally ordered set;
- upon termination one node is ‘leader’ and the rest are ‘lost’.

In this assignment we are going to look at two leader election algorithms on a ring topology:

- Part I: Chang-Roberts LEA or CR-LEA.
- Part II: Dolev-Klawe-Rodeh LEA or DKR-LEA (<http://w3.cs.huji.ac.il/~dolev/pubs/n-log-n.pdf>).

In each of these two algorithms, initially all processes are active. A process may become passive, in which case it is out of the race to become leader. The conditions under which it becomes passive depends on the algorithm and will be made precise below.

3 Part I: CR-LEA

This part of the assignment requests that you perform the task of implementing CR-LEA in Promela.

3.1 CR-LEA Description

The idea of the algorithm is that only the message with the highest id completes one round in the ring. Each node in the ring should behave as indicated by the following pseudocode:

```
1 send(id) to next process in ring;
2 while (true) do
3   receive (m);
4   if (m = id) then stop;    /* process is the leader */
5   if (m > id) then send(m); /* forward identifier, id becomes passive */
6   if (m < id) then do nothing; /* purge message */
7 od
```

Once the leader is elected, every node should be informed and print out the following message exactly once (per node): “I am node X and I know the leader is Y”.

Finally, make sure that the execution of your implementation does not end in a timeout.

Note: CR-LEA has message complexity $\mathcal{O}(N^2)$ (it takes at most $\frac{n(n+1)}{2}$ messages to elect a leader). DKR-LEA has message complexity $\mathcal{O}(N \log N)$.

3.2 Implementation

Implement CR-LEA in Promela. Nodes should be connected in a ring topology. The buffer size for the Promela channels should be set to $2 * N$ where N is the number of processes in the network.

Hint: It is strongly recommended that, on a first reading, you continue with the next section. The implementation for DKR-LEA will be provided for you, and will serve as a guideline for implementing, the more simple, CR-LEA.

4 Part II: DKR-LEA

This part of the assignment requests that you perform just one task, namely verify properties of DKR-LEA.

4.1 DKR-LEA Description


An implementation of DKR-LEA in Promela is provided for you below. First, an informal description. DKR adapts ideas from Franklin’s algorithm (which applies to undirected rings) with the aim of improving the message complexity of the Chang-Roberts algorithm. The basic idea depends on the fact that, since order of communication is preserved, the effect of the algorithm is as though the activities of the system could be separated into synchronous phases. Initially all

processes are active. In each phase, sufficient messages are passed so that each active process learns the current numbers of the two closest active processes on its left. With this information, each active process can determine if it is the active process immediately following a local maximum. If so, it assumes the local maximum as its own number, and continues as active on to the next phase. If not, it becomes passive and merely acts as a communication relay, passing on the messages initiated by active processes during later phases.

4.2 Implementation

Consider the following implementation¹ in Promela of DKR-LEA. It consists of two proctypes, `mnode` and `init`. The code for these proctypes are presented in two parts to ease readability. The former is the code that runs on each node in the ring; the latter initializes the ring.

It is suggested that you perform a symbolic execution of this algorithm in a ring with 5 nodes n_1 with id 1, ..., n_5 with id 5, connected as follows:



To simplify the state of the system you can write the messages that are exchanged on the arrows, the local state of each node next to the name of the node and you may assume that all nodes fire messages at the same time (although of course this is not necessarily the case in practice).

¹The code itself is from the spin website.

4.2.1 Part 1

```

1  #define N      5      /* number of processes in the ring */
2  #define L      10     /* size of buffer (>= 2*N) */
3  byte I;           /* will be used in init for assigning ids to nodes */
4
5  mtype = { one, two, winner }; /* symb. Msg. Names */
6  chan q[N] = [L] of { mtype, byte }; /* asynchronous channels */
7
8  proctype nnode (chan inp, out; byte mynumber)
9  {
10     bit Active = 1, know_winner = 0;
11     byte nr, maximum = mynumber, neighbourR;
12
13     xr inp; /* channel assertion: exclusive rcv access to channel in */
14     xs out; /* channel assertion: exclusive send access to channel out */
15
16     printf("MSC: %d\n", mynumber);
17     out!one(mynumber);
18 end:
19     do
20         :: inp?one(nr) ->
21             if
22                 :: Active ->
23                     if
24                         :: nr != maximum ->
25                             out!two(nr);
26                             neighbourR = nr
27                         :: else ->
28                             know_winner = 1;
29                             out!winner,nr;
30                     fi
31                 :: else ->
32                     out!one(nr)
33             fi
34         :: inp?two(nr) ->
35             if
36                 :: Active ->
37                     if
38                         :: neighbourR > nr && neighbourR > maximum ->
39                             maximum = neighbourR;
40                             out!one(neighbourR)
41                         :: else ->
42                             Active = 0
43                     fi
44                 :: else ->
45                     out!two(nr)
46             fi
47         :: inp?winner,nr ->
48             if
49                 :: nr != mynumber ->
50                     printf("MSC: LOST\n");
51                 :: else ->
52                     printf("MSC: LEADER\n");
53             fi;
54             if
55                 :: know_winner
56                 :: else -> out!winner,nr
57             fi;
58             break
59     od
60 }

```

4.2.2 Part 2

The `init` process assigns random ids to each node and then starts them.

```
1 init {
2   byte proc;
3   byte Ini[6];          /* N<=6 randomize the process numbers */
4   atomic {
5     I = 1; /* pick a number to be assigned 1..N */
6     do
7       :: I <= N ->
8         if /* non-deterministic choice */
9           :: Ini[0] == 0 && N >= 1 -> Ini[0] = I
10          :: Ini[1] == 0 && N >= 2 -> Ini[1] = I
11          :: Ini[2] == 0 && N >= 3 -> Ini[2] = I
12          :: Ini[3] == 0 && N >= 4 -> Ini[3] = I
13          :: Ini[4] == 0 && N >= 5 -> Ini[4] = I
14          :: Ini[5] == 0 && N >= 6 -> Ini[5] = I /* works for up to N=6 */
15        fi;
16        I++;
17      :: I > N -> /* assigned all numbers 1..N */
18        break
19    od;
20
21    /* start all nodes in the ring */
22    proc = 1;
23    do
24      :: proc <= N ->
25        run nnode (q[proc-1], q[proc%N], Ini[proc-1]);
26        proc++;
27      :: proc > N ->
28        break
29    od
30  }
31 }
```

4.3 Verification

1. Using assertions:
 - (a) Verify that if a node is selected as leader, then its id is the maximum one.
 - (b) Verify that only one leader is elected. Hint: add a variable `nr_leaders` to count the number of processes that think they are leaders.

5 Submission Instructions

Submit your Promela source code via Canvas by the indicated date and time. Submit a zip file named **Assignment5.zip** containing:

1. The Promela source for the first exercise
2. The Promela source for each verification item together with the output from spin (save as text).

In total you should have 4 files.

1. CR-LEA source code

2. CR-LEA running output as txt
3. DKR-LEA source code with assertion statements
4. DKR-LEA running output as txt