

Concurrent Programming

Exercise Booklet 9: Promela and Spin¹

Solutions to selected exercises (◇) are provided at the end of this document. Important: You should first try solving them before looking at the solutions. You will otherwise learn **nothing**.

1 Basic Promela

Exercise 1. Do all interleavings in the following program ensure that x is always between 0 and 200?

Encode it in Promela and use Spin to determine the answer.

```
x = 0;
Thread.start { // P      Thread.start { // Q      Thread.start { // R
  while (true) {          while (true) {          while (true) {
    if (x < 200)            if (x > 0)                if (x == 200)
      x++;                x--;                    x = 0;
    }
  }
}
```

Exercise 2. (◇) Implement the following entry/exit protocol (Attempt I) seen in class, in Promela. Use the Promela code in the slides as an aid in understanding Promela syntax.

```
int turn = 1;
Thread.start { // P      Thread.start { // Q
  while (true) {          while (true) {
    await (turn == 1);    await (turn == 2);
    turn = 2;            turn = 1;
  }
}
```

Exercise 3. Implement the following entry/exit protocol (Attempt III) seen in class, in Promela. Use the Promela code in the slides as an aid in understanding Promela syntax.

```
1 boolean wantP = false;
2 boolean wantQ = false;
3 Thread.start { //P      Thread.start { // Q
4   while (true) {          while (true) {
5     // non-critical section    // non-critical section
6     wantP = true;            wantQ = true;
7     await (!wantQ);          await (!wantP);
8     // CRITICAL SECTION      // CRITICAL SECTION
9     wantP = false;           wantQ = false;
10    // non-critical section    // non-critical section
11  }
12 }
```

Exercise 4.

Draw the transition system of the following two programs separately and then compare them:

¹Sources include: <http://www.cs.toronto.edu/~chchik/courses01/csc2108/lectures/spin.2up.pdf>

```

1 % Program 1
2 byte state = 1;
3 active proctype A(){
4     atomic {
5         (state==1) ->
6         state = state+1
7     }
8 }
9 active proctype B() {
10     atomic {
11         (state==1) ->
12         state = state-1
13     }
14 }

```

```

1 % Program 2
2 byte state = 1;
3 active proctype A(){
4     (state==1) ->
5     state = state+1
6 }
7 active proctype B() {
8     (state==1) ->
9     state = state-1
10 }

```

Exercise 5. Check whether the following algorithm guarantees mutual exclusion by adding an auxiliary variable `critical` and appropriate statements. Do you recognize this algorithm?

```

1 bool flag[2]
2 bool turn
3
4 active [2] proctype user()
5 {
6     flag[_pid] = true
7     turn = _pid
8     (flag[1-_pid] == false || turn == 1-_pid)
9
10 crit:    skip    // critical section
11
12     flag[_pid] = false
13 }

```

Exercise 6. What happens in the previous algorithm if you exchange the line `flag[_pid]= true` with the line `turn = _pid`?

Exercise 7. Consider the following simplified presentation of the Bakery Algorithm for two processes:

```

1 int np,nq =0;
2 Thread.start { // P
3     while (true) {
4         // non-critical section
5         np = nq + 1;
6         await nq==0 or np<=nq;
7         // CRITICAL SECTION
8         np = 0;
9         // non-critical section
10    }
11 }

```

```

1 Thread.start { // Q
2     while (true) {
3         // non-critical section
4         nq = np + 1;
5         await np==0 or nq<np;
6         // CRITICAL SECTION
7         nq = 0;
8         // non-critical section
9     }
10 }
11 }

```

1. Encode it in Promela.
2. We mentioned in the lectures that if assignment is not atomic, then mutual exclusion is not guaranteed. Verify this in spin. Since assignment is atomic in spin, you first must split line 5 (in P and Q) into two assignment operations.

3. Add the command `np = 1` in thread P just before line 4 and add `nq=1` in thread Q just before line 4. Show, using Spin, that the resulting program does enjoy mutual exclusion.
4. If you verify the code above (for mutex), where assignment is assumed atomic, Spin seems to report an assertion violation. Why? Hint: run the following code and bear in mind that the `byte` type has a range of 256 different values:

```

1 byte n=0;
2
3 active proctype P() {
4     int i=0;
5     for (i: 1..1000) {
6         n++;
7     }
8 }

```

Exercise 8. The following extension to 3 processes of Dekker's algorithm is known to deadlock.

```

1 int turn=0
2 flags = [false, false, false]
3
4 3.times {
5     int id = it
6     Thread.start {
7         int left = (id+2)%3;
8         int right = (id+1)%3;
9         while (true) {
10            flags[id] = true;
11            while (flags[left] || flags[right])
12                if (turn == left) {
13                    flags[id] = false;
14                    await (turn==id);
15                    flags[id] = true;
16                }
17            }
18            // Critical Section
19            turn = right;
20            flags[id]=false;
21        }
22    }

```

Encode it in Promela and verify where it deadlocks (there are multiple traces that lead to deadlock, exhibiting one suffices). Here is some code that you can start from.

```

1 byte turn;
2 bool flags[3];
3
4
5 proctype P() {
6     byte myId = _pid-1;
7     /* complete here */
8 }
9
10 init {
11     turn=0;
12     byte i;
13     for(i:0..2) { flags[i] = false; }
14     atomic {
15         for (i:0..2) { run P(); }

```

```

16 }
17 }

```

Exercise 9. (\diamond)

1. Define a general semaphore with operations `acquire(sem)` and `release(sem)`, where `sem` is a numeric shared global variable. Since there are no functions in Promela, you can declare macros using `inline`. Hint: use `atomic` and blocking expressions.
2. Check that your solution is correct by providing an example of its use.

Exercise 10. Implement the Bar exercise from eb5, in Promela, using the semaphore encodings from Exercise. 9. Use Spin to show that the required invariant is upheld, i.e. that only one Jets fan goes in for every two Patriot fans, by inserting an appropriate assertion. In your example use 20 Jets fans and 20 Patriot fans. Here is a stub:

```

1 byte ticket = 0;
2 byte mutex = 1;
3 /* additional declarations here */
4
5
6
7
8 inline acquire(sem) {
9     atomic {
10         sem>0 -> sem--
11     }
12 }
13
14 inline release(sem) {
15     sem++
16 }
17
18 active [20] proctype Jets() {
19     /* complete */
20 }
21
22
23 active [20] proctype Patriots() {
24     /* complete */
25 }

```

Exercise 11. Implement the Bar exercise from eb6 (monitors!), in Promela, using the semaphore encodings from Exercise. 9. Use Spin to show that the required invariant is upheld, i.e. that only one Jets fan goes in for every two Patriot fans, by inserting an appropriate assertion. Here is a stub:

```

1 int N=4;
2
3
4 inline jets() {
5     // complete
6 }
7
8 inline patriots() {
9     // complete
10 }

```

```

11
12 proctype Jets() {
13     jets();
14 }
15
16
17 proctype Patriots() {
18     patriots();
19 }
20
21
22 init {
23     int i;
24     for (i: 1 .. N) {
25         atomic {
26             run Jets();
27             run Patriots()
28         }
29     }
30 }

```

Exercise 12. Implement the Car Wash exercise from eb5, in Promela, using the semaphore encodings from Exercise. 9. Use Spin to show that there is at most one car in any of the three stations. Here is a stub:

```

1  #define N 3 /* Number of Washing Machines */
2  #define C 10 /* Number of Washing Machines */
3
4  byte permToProcess[N]
5  byte doneProcessing[N]
6  byte station0 = 1
7  byte station1 = 1
8  byte station2 = 1
9
10
11 inline acquire(s) {
12     atomic {
13         s>0 -> s--
14     }
15 }
16
17 inline release(s) {
18     s++
19 }
20
21 proctype Car() {
22     /* complete */
23 }
24
25
26 proctype Machine(int i) {
27     /* complete */
28 }
29
30 }
31
32 init {
33     byte i;
34

```

```

35   for (i:0..(N-1)) {
36       permToProcess[i]=0;
37       doneProcessing[i]=0;
38   }
39
40   atomic {
41       for (i:1..(C)) {
42           run Car();
43       }
44       for (i:0..(N-1)) {
45           run Machine(i);
46       }
47   }
48 }

```

2 Channels

Exercise 13. (\diamond) What does the following program print?

```

1  proctype A(chan q1) {
2      chan q2;
3      q1?q2;
4      q2!123
5  }
6  proctype B(chan qforb) {
7      int x;
8      qforb?x;
9      printf("x=%d\n",x)
10 }
11 init {
12     chan qname = [1] of { chan };
13     chan qforb = [1] of { int };
14     run A(qname);
15     run B(qforb);
16     qname!qforb
17 }

```

Exercise 14. (\diamond) Implement a binary semaphore in Promela using message passing and synchronous communication. You should have two proctypes, `semaphore` and `user`. Here is the code for the user:

```

1  #define acquire 0
2  #define release 1
3  chan sema = [0] of { bit }; /* synchronous channel */
4  proctype semaphore() {
5      /* complete this */
6  }
7  proctype user() {
8      do
9          :: sema?acquire;
10         /* crit. sect */
11         sema!release;
12         /* non-crit. sect. */
13     od
14 }
15 init {

```

```

16  run semaphore();
17  run user();
18  run user();
19  }

```

Exercise 15. Implement the turnstile example from class using channels. Here is some code to get you started:

```

1  mtype { bump , read };
2  chan counter = [0] of { mtype, chan }
3  chan reply[2] = [0] of { byte }
4
5  proctype Counter() {
6      /* complete */
7  }
8
9  proctype Turnstile() {
10     /* complete */
11 }
12
13 init {
14     byte x;
15     atomic {
16         run Counter();
17         run Turnstile();
18         run Turnstile();
19     }
20     _nr_pr==2;
21     counter!read(reply[0]);
22     reply[0]?x;
23     printf("%d\n",x) /* should print 20 */
24 }

```

3 Solutions to Selected Exercises

Answer to exercise 2

```

1  byte turn=1;
2
3  active proctype P() {
4      do
5          :: do
6              :: turn==1 -> break;
7              :: else
8                  od;
9              printf("P went in \n");
10             turn = 2
11         od
12     }
13
14     active proctype Q() {
15         do
16             :: do
17                 :: turn==2 -> break;
18                 :: else
19                     od;
20                 printf("Q went in \n");

```

```

21     turn = 1
22     od
23 }

```

The following variation is acceptable but not quite right since the “await” does busy-waiting, it does not block:

```

1  byte turn=1;
2
3  active proctype P() {
4      do
5          :: turn==1;
6              printf("P went in \n");
7              turn = 2
8      od
9  }
10
11 active proctype Q() {
12     do
13         :: turn==2;
14             printf("Q went in \n");
15             turn = 1
16         od
17 }

```

Answer to exercise 9

```

1  inline acquire(sem) {
2      atomic {
3          sem>0;
4          sem--
5      }
6  }
7  inline release(sem) {
8      sem++
9  }

```

Answer to exercise 13

- Init sends qforb to A on qname
- A sends 123 to qforb
- B receives 123 on qforb and prints it

Answer to exercise 14

```

1  #define acquire 0
2  #define release 1
3  chan sema = [0] of { bit };
4  proctype semaphore() {
5      byte count = 1;
6      do
7          :: (count == 1) -> sema!acquire; count = 0
8          :: (count == 0) -> sema?release; count = 1
9      od
10 }
11
12 proctype user() {
13     do

```



```
14     :: sema?acquire;  
15     /* crit. sect */  
16     sema!release;  
17     /* non-crit. sect. */  
18   od  
19 }  
20 init {  
21   run semaphore();  
22   run user();  
23   run user();  
24 }
```